
Subject: Re: Getting the current color table
Posted by [JD Smith](#) on Tue, 06 Dec 2005 21:55:19 GMT
[View Forum Message](#) <> [Reply to Message](#)

On Tue, 06 Dec 2005 10:44:05 -0800, inaki.ugarte wrote:

> Hi,
> is there any recent development in this previously asked question? How do
> I get the color table number used by the last call to LOADCT? I would
> prefer to do it in a general way, without having to call external
> routines.
> Cheers,
> Inaki

I think at some point many years ago this bugged me enough that I wrote a tool called cload which stores the color table, and the color min/max/gamma as well, in a custom system variable. I don't use it anymore, but it might be of interest. If someone else uses LOADCT in the meantime, you'd be out of luck.

```
;; load with loadct, but set the !ctabl system variable too.
pro cload, table_number, NO_SET=ns, _REF_EXTRA=e
  if n_elements(table_number) ne 0 then begin
    loadct, table_number, /SILENT, _EXTRA=e
    !ctabl.cmap=table_number
    if NOT keyword_set(ns) then begin
      !ctabl.high=1.
      !ctabl.low=0.
      !ctabl.gamma=1.
    endif
  endif else loadct, _EXTRA=e
end
```

you have to setup that sysvar first:

```
pro cdef, _EXTRA=e
  defsysv, '!ctabl', Exists=exi
  if exi eq 1 then return
  ;; we need to define it...
  ;; make and delete window to get the *actual* number of colors used
  window, 1, xsize=1, ysize=1, /PIXMAP, _EXTRA=e
  wdelete
  defsysv, '!ctabl', {!ctabl, cmap:0, low:0., high:1., gamma:1.0}
  return
end
```
